
1 はじめに
バンド計算は通常逆格子空間（波数空間）の適当な代表点を選んで実行される。このとき選ばれる代表点
の数は数十から数百が選ばれる。しかしこれでは金属的な物質でフェルミ面を作図したり、熱起電力などの
伝導に関する量の計算に十分な密度ではない。たとえばこのホームページに紹介した TPERSP の説明に使
われたフェルミ面の作図には１５万点を超える波数でのエネルギー値が使用されている。ここで紹介する
プログラム TSBDFT は、あたえられた対称性に基づいて、任意の波数ヴェクトルでのバンドエネルギー
をあたえる解析的な関数を求め、それを用いて必要な密度でバンドエネルギーを計算する。TSBDFT では
TSPACE の機能を使っている。ときどき説明に使われる専門用語に戸惑われる方は、たとえば、拙著「ブ
リルアンゾーンとは」(丸善)などを読んでいただきたい。また「空間群のプログラム TSPACE」の３、４
章で説明している TSPACE が使用している実格子空間、逆格子空間での座標系、単位系を理解した上で、
この説明を読んでいただきたい。TSBDFT では、TSPACE の上で使用するように設計されている。した
がって「空間群のプログラム TSPACE」の記述と重複する部分が表れる。その場合には TSP:Pxxx のよう
にして引用する。

TSBDFT が使用する逆格子空間の代表点は、TSPACE が想定する逆格子空間の座標軸 a∗, b∗, c∗ 軸に
沿って、原点から、その方向の再近接の逆格子点までを等間隔に刻んだ３次元格子の上に採られていること
を期待している。これらの点は、必ずしもブリルアンゾーンの中に含まれている必要はない。ただし、互
いに同等な点を含んでいると誤動作の原因になる。当然なことであるが、３次元の３方向で、間隔をでき
るだけ等しくとってある必要がある。例えば c 軸が長く c∗ が短い結晶の場合は c∗ 軸方向の刻みの数より
a∗ 軸方向の刻みを多くとって、互いの間隔が近くなるようにする必要がある。TSPACE にはサブルーチン
TSKPGN が用意されているのでこの利用が便利である。TSKPGN の出力では、k = 0 の Γ点が、最初に
くるようになっているが TSBDFT のためには、このようになっているのが望ましい。
逆格子空間の代表点の選び方に、 a∗, b∗, c∗ 軸に沿うのではなく、基本逆格子ベクトルに沿って刻む方法
もあるが、TSBDFT のためには、この方法は適当ではない。最も分かりやすい例は底心直方格子の場合で
ある。二つの a∗, b∗ 面内の基本逆格子ベクトルは、長さが等しくどちらも a∗, b∗ 軸の間になっている。した
がって基本逆格子ベクトルに沿って刻むと、a∗, b∗ の長さに関係なく a∗, b∗ 方向の刻みが同じになってしま
う。この二つの軸の長さの違いが２倍以上の物質が実際に数多く存在している。このような代表点の取り方
は TSBDFT のためだけではなく、密度汎関数理論に則った、電荷密度の計算のためにも望ましくはない。
基本格子ベクトルや基本逆格子ベクトルを使う議論は、式の上では簡潔になり便利であるが、我々のあま
り得意でない３次元感覚をさらに狂わせてしまう欠点を持っている。基本格子ベクトルで表した係数には、
そのベクトルが持っている対称性が直接には反映されていないからである。これが、実用的なプログラムを
目指す TSPACE が、a, b, c、a∗, b∗, c∗ を基準にした、座標系に固執している理由の一つである。
以下のプログラムの説明では、ソースプログラムをアルゴリズムの鍵になる部分以外は、一々引用しない

ことが多いので各自でソースを手元に持ちながら読んでいただきたい。

2 内挿関数
バンドエネルギー E(k) は波数空間での周期性がある。

　　 E(k + G) = E(k)　 (1)

ここで G は任意の逆格子ベクトルである。任意の実格子ベクトルを Tn として exp(iTn · k) はこの周期性
を持っている。
さらに、結晶が回転の対称性を持つ場合には E(k) もやはり同じ回転の対称性を持つ。

　　　 E(k) = E(αk) (2)

らせん、映進のように、結晶の周期でない並進を伴っている場合でも同じように成り立つ。
空間群 G を、その並進群 T を法として剰余類に

　　 G = (ε|0)T + (α|uα)T + (β|uβ)T + · · · (3)

と分類すると　 ε, α, β, · · · は一つの結晶点群 GP になる。そこで
∑

α∈GP

exp(iαTn · k) (4)

とすれば、 ∑

α∈GP

exp(iαTn · βk) =
∑

α∈GP

exp(iβ−1αTn · k) (5)

だから、回転の対称性を持った関数になる（組み換え定理）TSP:p9。
バンドエネルギー E(k) にはもう一つの重要な対称性がある。それは時間反転の対称性に起因している。
結晶が空間反転の対称性を持っていれば、当然

E(k) = E(−k) (6)

となるが、たとえ空間反転の対称性がない場合でも、このことを時間反転の対称性が導く。スピン軌道相互
作用を含めない議論では、これで議論を終わりにしてよい。強磁性状態などで、スピンの縮退が解けていて
時間反転の対称性が壊れている場合でも、空間座標とスピン座標が互いに関係していないので、空間座標
の部分では時間反転の対称性は保たれているからである。もちろん時間反転で生じる、ペアリング、ダブリ
ングの縮退度の変化は当然考慮にいれてのことである。
スピン軌道相互作用を含めると議論が次のように変わる。磁性状態では時間反転の対称性は壊れているの
で、ここでは扱わない。空間反転がある結晶では、時間反転と空間反転の両方が E(k) = E(−k) となるよ
うに働く。このとき時間反転はスピンの向きを反転させるのに対し、空間反転はスピンの向きを変えないの
で、結果的にスピンの自由度に起因する二重縮退がすべての k で残ることになる。空間反転を持たない結
晶では、時間反転だけが E(k) = E(−k) となるように働くため、一般の k ではこの二重縮退が残らない。
とにかく E(k) = E(−k) は保証されているので、

∑

α∈GP

cos(αTn · k) (7)

は E(k) を表す基底として使用できる。
しかし、実際の E(k) は k の関数として連続ではあるが、その微分は連続ではない。(7) は微分が連続で
あるから、このことを表現できない欠点を持っている。微分が不連続になるのはつぎの２つの場合が挙げら
れる

1. 対称軸、対称面での偶然縮退

2. 一般の点から対称面、対称面から対称軸、対称軸から対称点などの対称が変わる点での接続

これに対処する方法はいろいろ考えられるが、ここで紹介する TSBDFT では、このことの手当はしてい
ない。汎用的に使用できるようにすることを優先したからである。この内挿法を使えば、TPERSP の応用
でフェルミ面を描くことが、ほとんどすべての結晶で可能になる。描かれたフェルミ面に不自然な点があれ
ば、代表点の数を増やして変化がないかどうか確かめる必要がある。特に偶然縮退がフェルミ面の近くで起
きていれば、(7) が、偶然縮退の不自然な変化をカバーしていること確かめるなどの注意を払うべきである。
AYBAND を使って該当する箇所を JOPT=1 にしてプロットしてみるのもおすすめである。AYBAND で
は上に揚げた二つの問題点はカバーされている。AYBAND の JOPT=1 の図は、使われている代表点の位
置が図面に現れるので、特異な形のフェルミ面が本物かどうかの判定に役に立つ。とにかく、もっともらし
く見える結果の真偽のほどを保証するのは使用者個人の責任である。

3 プログラムの流れ
プログラム 3.1 は TSBDFT のメインプログラムの全体である。プログラム 3.2 は 最初の SUBROU-

TINE TSPPRP で、プログラム 3.3は 二番目の SUBROUTINE FITPRM である。ENRRED、SSSLAT、
FNNMAT、FUEINT については、節を変えて説明する。

プログラム 3.1
C--------1---------2---------3---------4---------5---------6---------7---------8

C Main for fitting of Bands energies and components

CALL TSPPRP

WRITE(6,*) ’ End of TSPPRP’

CALL FITPRM

WRITE(6,*) ’ End of FITPRM’

CALL ENRRED

WRITE(6,*) ’ End of ENRRED’

CALL SSSLAT

WRITE(6,*) ’ End of SSSLAT’

CALL FNNMAT

WRITE(6,*) ’ End of FNNMAT’

CALL FUEINT

WRITE(6,*) ’ End of FUEINT’

STOP

END

プログラム 3.2
C--------1---------2---------3---------4---------5---------6---------7---------8

SUBROUTINE TSPPRP

C***********************************************************************

C PREPERATION TO CALL TSPACE *

C***********************************************************************

IMPLICIT REAL*8 (A-H,O-Z)

COMMON/SCL/NATOM,NKAT,IPR,ISO,E0,DE,NNEE,NELEC,ENEND

DIMENSION JB(2,3)

READ(1,*)

READ(1,*) IL,NGEN,INV

CALL TSPACE(IL)

DO 1 I=1,NGEN

READ(1,*) JA,((JB(J,K),J=1,2),K=1,3)

CALL TSGENR(JA,JB)

1 CONTINUE

CALL TSPGRP(INV)

IF(IPR.GE.3) CALL TSPGDS

READ(1, ＊) A,B,C

READ(1, ＊) CA,CB,CC

CALL TSLATC(A,B,C,CA,CB,CC)

READ(1,*) NATOM,NKAT

RETURN

END

fort.1 からデータを読み込んで格子の型 IL と生成元をあたえて、空間群を決定したあと、格子定数をあた
えている。最後の READ(1,＊) NATOM,NKAT は、当面必要がないが、4 節で説明する s, p, d, f 成分の
読み込み用である。このプログラムは　 TSP:p71 p73 の例 5.2,5.3とほとんど同じである。TSPACE が使
用している空間群、格子定数のあたえ方の詳細はこの部分を参照されたい。
プログラム 3.3
C--------1---------2---------3---------4---------5---------6---------7---------8

SUBROUTINE FITPRM

IMPLICIT REAL*8 (A-H,O-Z)

INCLUDE ’prmfit.f’

COMMON/SPG2/IL,NG,IG(48),JV(2,3,48)

COMMON/NXYZ/NX,NY,NZ,N64,JS

CHARACTER*4 CHARA4

CHARACTER*10 MAGNET

C

READ(5,150) MAGNET

150 format(A10)

CHARA4=MAGNET

IF(CHARA4.EQ.’NONM’) ISO=1

IF(CHARA4.EQ.’MAGN’) ISO=2

IF(CHARA4.EQ.’SPIN’) ISO=3

NDFAC=1

IF(ISO.EQ.3) THEN

DO 10 I=1,NG

IF(IL.GE.1.AND.IG(I).EQ.25

& .OR.IL.LE.0.AND.IG(I).EQ.13) NDFAC=2

10 CONTINUE

END IF

IF(ISO.EQ.3.AND.NDFAC.EQ.1) ISO=4

write(6,*) ’NDFAC=’,NDFAC,’ISO CODE=’,ISO

READ(5,*) NELEC,NB1,NB2

READ(5,*) AM

READ(5,*) IPR

READ(5,*) NX,NY,NZ,N64,JS

NBLOCK=1

RETURN

END

このサブルーチンは標準入力 (fort.5)から読み込んで、内挿プログラムの仕様を決める。最初に CHAR-
ACTER*10 変数 MAGNET に非磁性 (NONMAGNETIC)、強磁性 (MAGNETIC)、SPIN軌道相互作用
を含めた計算 (SPIN-ORBIT) を読み込んで、入力されるバンド計算がこのいずれかであるかを指定する。
2 節で述べたようにスピン軌道相互作用を含めた場合には、空間反転のあるなしが E(k) の縮退に影響する
ので、ここで、それがある場合を ISO=3 , ない場合を ISO=4 として区別している。
つぎの READ 文の NELEC は計算されたバンドに含まれるはずの電子数をあたえるが、TSBDFT では
結果として使用しないので、なにがあたえられても結果に影響しない。バンド計算のプログラムが原子番号
等から自動的にあたえてくれているのを、使用者が再認識するようにこの入力を消さずに置いてある。内
挿は NB1番目のバンドから NB2番目のバンドまで実行される。常磁性の物質ではフェルミ面があるバン
ドをカバーするように選べばよい。半導体や絶縁体ではバンドギャップを挟む適当な数のバンドを選ぶ。

AM は SSSLAT で用意する格子ベクトルの大きさの範囲をあたえる。IPR はプログラムの出力制御用で
ある。この TSBDFT はあたえられた k-点の数が増加すると、必要なベースの数が増え、計算時間も長く
なる。IPR の値としては 0,3,4,5 が現在のバージョンには使われている。適当に選んで計算の進行をトレー
スすることができる。NX,NY,NZ,N64,JS については、それらが使用されるサブルーチン FUEINT の説明
の中でふれる。

4 バンド計算の結果の読みとり:ENRRED

プログラム 4.1はサブルーチン ENEREDの中で、バンド計算の結果が出力されているファイルから読み込
んでいる部分の抜粋である。われわれが使っている FLAPWの出力に合わせてある。もちろん他の方法で計算
されたものでも、この形式に合わせてあればTSBDFTの入力として使用できる。40 READ(2,150,END=41)
の FORTAN の手法でファイルの終わりを検出しているので、データが何組あるかは、指定しなくてもよ
い。各 ~k 毎に、またその ~k での既約表現ごとにエネルギーの低いものから順に並んでいることを期待して
いる。
プログラム 4.1

40 READ(2,150,END=41) KK,(KX(J),J=1,3),IC,IUD,MRR,MRN,NDEG,NST,NEIG

IF(NEIG.GT.MAXEIG) THEN

WRITE(6,*) ’ MAXEIG=’,MAXEIG,’ NEIG=’,NEIG

STOP

END IF

IF(NEIG.GT.0) READ(2,*) (EIG(J),J=1,NEIG)

150 FORMAT(I3,2X,3I3,1X,2I3,2X,A2,I2,3I3)

READ 文の並びに現れた変数の意味は

• KK — バンド計算のときに ~k につけられた通し番号、TSBDFT では読み捨てる。

• KX(3) — ~k の座標の分子。

• IC — ~k の座標の分母。

• IUD — スピンの UP(1) or DOWN(2)、パラのときには１。

• MRR — ~k の名前、TSBDFT では用いない。

• MRN — 既約表現の番号、TSBDFT では使用しない。

• NDEG — 縮退度、TSBDFT ではこれを元に既約表現ごとにあたえられたエネルギーを、大きさの
順番に並べ直している。なお、プログラム 3.3で ISO=3 とした場合（スピン軌道相互作用を含めて
いて、空間反転を持つ場合）は NDEG は偶数になっているので、確かめた上で NDEG を半分にし
ている。

• NST — ~k の星の数、TSBDFT では用いない。

• NEIG — その既約表現に属する、状態の数。

である。変数の型は FORMAT から分かるように MRR 以外はすべて整数型である。MRR は２文字の文
字型変数であるが、上記のように用いないのでなにが入っていてもよい。もちろんスペースでもよい。

入力データ 4.1
45 4 4 4 8 1 R 1 4 1 5

-1.90115 -.46922 .19579 .53289 .90794

.3088 .1071 .0047 .0005

.0261 .2785 .0037 .0005

.0034 .2197 .0018 .0002

.0343 .0682 .0015 .0001

.0061 .0432 .0038 .0002

45 4 4 4 8 1 R 2 4 1 4

-.73549 -.57574 .45021 .88001

.2398 .0529 .0002 .0000

.0059 .2323 .0024 .0004

.0283 .0528 .0014 .0001

.0050 .0064 .0014 .0000

45 4 4 4 8 1 R 5 4 1 2

.22001 .93867

.0000 .2484 .0020 .0003

.0000 .0528 .0028 .0001

45 4 4 4 8 1 R 6 4 1 2

-.56316 1.06384

.0000 .2631 .0028 .0004

.0000 .0427 .0015 .0001

入力データ 4.1は窒素の分子結晶に対する、バンド計算結果の R 点の部分である。ここでのｋ点群は位
数が 24の Th で２次元ばかり６個の既約表現を持っている。R1, R3 と R2, R4 のペアーがそれぞれ時間反
転でペアリングをして４重縮退になる。また R5 と R6 は時間反転でダブリングをしている。TSBDFT は
時間反転でペアリング、ダブリングが起きた場合にはこの例のように扱われていることを期待している。　

1. ペアリングでは番号の若い方の既約表現の縮退度を２倍して、そのペアーの相手の既約表現は出力に
入れない。

2. ダブリングでは縮退度を２倍して、同じエネルギーを二つ並べて出力することはしない。

この入力データにはエネルギーに続いて、そのエネルギーを持つ状態の性格を示す数値がそれぞれ４個
づつついているが、TSBDFT はこの部分も読みこんで格納している。これは、各状態の、それぞれの原子
球内での s, p, d, f 成分である。将来この性格を示す数値も合わせて内挿できるようにする予定である。
サブルーチン ENRRED は fort.2 からの入力データを整理して

プログラム 4.2
COMMON/ENR/KXM(3,MAXKPT),ICM(MAXKPT),IUDM(MAXKPT)

& ,NEIGM(MAXKPT),EIGM(MAXEIG,MAXKPT),NKPINT,NBAND,NB1,NB2

に格納した後、バンド数 NBAND を各 k でのエネルギー値の数の最小値にして、もし NB2 がこれを超え
ていれば、NB2=NBAND としている。このサブルーチンは最後に NBAND,NB1,NB2 を標準出力に出し
ている。NKPINT は、代表点の数であるが、強磁性状態の場合には UP と DOWN を別のデータとして
扱っているので、実際の代表点の数の２倍になっている。

5 内挿基底関数の準備:SSSLAT

SSSLAT では、TSPACE が内部で使用している隠しサブルーチンや内部変数を直接使用している。プロ
グラムの流れの説明のなかで触れていくことにする。
サブルーチン SSSLAT の目的は 式 (7)の基底関数を組み上げることである。結果が次の COMMON 領
域に格納される。
プログラム 5.1

COMMON/LATVC/LATOM(3,MAXLP),ALM(MAXLP),MP(MAXLP),NLAT,NSHEL,ICD

最後の ICD は格子ベクトルの共通分母で、三方格子なら 3、底心 、面心、体心格子 なら 2 で、他の単
純格子なら 1 になっている。

NSHEL は用意された基底関数の数、NLAT はそれらに含まれる格子ベクトルの総数である。
LATOM(3,MAXLP) には、NLAT 個の格子ベクトルが並べてあり、その大きさが ALM(MAXLP) であ
る。この中には Tn = −Tm の関係になるような格子ベクトルは含まれないようにしてある。MP(MAXLP)
は、基底関数を構成している最後の格子ベクトルの LATOM 上の番号が入っている。
このサブルーチンの流れは次のようである。

1. まず格子の型によって、TSP:p62～63の (5.2)～(5.6) にしたがって基本格子ベクトルをあたえたの
ち、その整数倍の線形結合で順次格子ベクトル作って、LATOM に格納していく。このとき大きさ
が FITPRM であたえた AM を超えるものを取り除きながら、大きさの順番にソートしておく。格
子ベクトルの大きさの計算には、TSPACE の隠しサブルーチン ZZZY43(V,DC,S) が使われている。
TSP:(5.2)～(5.6) は基本格子ベクトルを a, b, c を基準にした座標系であたえているので、使用してい
る座標系はやはり、a, b, c を基準にしたものである。

基本格子ベクトルの整数倍で格子ベクトルを生成するアルゴリズムはごく自然に見えるが、実際的に
は問題を含んでいる。たとえば、三方格子で、c軸が長い場合は、基本格子ベクトルは互いに接近して
いる。したがって c軸に垂直な方向の格子ベクトルを表すには、絶対値が大きな整数が必要になる。
c軸が短いときには逆に基本格子ベクトルは互いに開いており、c軸の方向の格子ベクトルを表すの
に大きな整数を必要とする。同様なことは格子定数が互いに大きく異なる体心直方格子のような場合
にもおきる。サブルーチン SSSLAT では探索する整数の範囲を AM を基本格子ベクトルの大きさで
割った値の５倍と一応大きめに設定している。しかしこれでも不十分なこともありうるので、その場
合は ’ STOP AT 26’ とコメントを出してプログラムを止めるようにしてある。これは、ある格子ベ
クトルを種にして、それと回転の対称性で結びつく格子ベクトルを探している箇所で該当する格子ベ
クトルが見つからない場合に相当している。a, b, c を基底にすると、1/2 とか 1/3 が現れて泥臭く見
えるかも知れないが、このほうが明らかにわれわれの直感と結びついている。

2. 次のステップの最初に大きさの等しいものごとにグループに分ける。

DO 52 II=2,NLAT ～ 52 CONTINUE

このグループの数が MSHEL で、境目が ML(MAXLP) である。

3. 次にこの大きさの等しい格子ベクトルを、(7)式を構成するように、さらにグループ分けする。

DO 81 IE=2,MSHEL ～　 81 CONTINUE の長い手続きになる。(7)式を構成する格子ベクトルは、
基底の間で重なることはない。配列 IFLA(MAXLP) は、各格子ベクトルにつけた目印であり、同じ
基底 (7)式に含まれる格子ベクトルのそれは、互いに等しくなるようにしてある。

まだ使われていない格子ベクトルの IFLA は 0 にしてあるので、それを見つけてその番号を ISAR
とする。DO 72 JFLA=1,50 に現れた、50 は同じ大きさになる格子ベクトルを (7)式を構成するよう
にグループ分けしたときの最大の組数を大きめに見積もった値である。FCC で Oh の点群の場合で
3281 個の代表点に対応して 10302 組のベースを作った場合に 27 組が表れたのが最大である。次節
で述べるように用意すべきベースの数は代表点の数の２倍から３倍が必要である。全部使われておれ
ば、73 CONTINUE の外に出る。この場合は DO 72 のループの外に出てもよいが、GO TO 72 と
しているのは、単に文番号の数を節約しているだけである。

プログラム 5.2
DO 72 JFLA=1,12

IF(JFLA.EQ.1) THEN

ISAR=ISTA

ELSE

DO 73 I1=ISTA,IEND

IF(IFLA(I1).EQ.0) GO TO 74

73 CONTINUE

GO TO 72

74 ISAR=I1

END IF

4. 上で決めた、たねの格子ベクトルを LATF(3)として、これを空間群の回転要素で回転して、LATT(3)
とする。回転要素のコード番号は TSPACE の COMMON/SPG2/ の IG(48) に格納されていて、そ
の数は NG 個である。また、このコード番号の回転操作は COMMON/SPG1/ の IT(3,48) に入って
いる。TSP:P7～11 のコード表で、X,Y,Z,W としてあるところが IT では、それぞれ 1,2,3,4 になっ
ている。またたとえば、-X は -1 に -W は -4 のようにしてある。

LATT(3) か -LATT(3) は当然どこかにあるはずだから、見つけて JFLA を目印につけておく。もし
みつからなければ、異常だから STOP が働く。

プログラム 5.3
LATF(1)=LATOM(1,ISAR)

LATF(2)=LATOM(2,ISAR)

LATF(3)=LATOM(3,ISAR)

IFLA(ISAR)=JFLA

DO 22 I1=2,NG

DO 23 K=1,3

IF(IABS(IT(K,IG(I1))).EQ.4) THEN

LATT(K)=LATF(1)-LATF(2)

IF(IT(K,IG(I1)).LT.0) LATT(K)=-LATT(K)

ELSE

M=IABS(IT(K,IG(I1)))

LATT(K)=LATF(M)

IF(IT(K,IG(I1)).LT.0) LATT(K)=-LATT(K)

ENDIF

23 CONTINUE

DO 26 ID=1,2

DO 24 I2=ISAR,IEND

IF(IFLA(I2).NE.0.AND.IFLA(I2).LT.JFLA) GO TO 24

DO 25 K=1,3

IF(LATT(K).NE.LATOM(K,I2)) GO TO 24

25 CONTINUE

IF(IFLA(I2).EQ.0) IFLA(I2)=JFLA

GO TO 22

24 CONTINUE

LATT(1)=-LATT(1)

LATT(2)=-LATT(2)

LATT(3)=-LATT(3)

26 CONTINUE

WRITE(6,*) IE,I1,ISTA,IEND,ISAR,LATF,LATT,’ STOP AT 26’

STOP

22 CONTINUE

72 CONTINUE

5. 最後の手続きは同じ目印のものを集めることで、また DO 78 JFLA=1,50 のループで行っている。

新しい結晶で使用するときに、IPR を 4 にして一度出力を見ておくのも無駄でないかもしれない。

6 内挿関数の係数を決定:FNNMAT

サブルーチンの最初に、kー空間での真の代表点の数を NFSHEL としている。NKPINT は、代表点の
数であるが、強磁性状態の場合には UP と DOWN を別のデータとして扱っているので、実際の代表点の
数の２倍になっている。4 節の ENRRED であたえられている、逆格子空間の代表点 k に対して、5 節で
求めた基底関数 ∑

α∈GP

cos(αTn · k) (8)

の値を計算していく。基底関数を構成する格子ベクトルの組は、最後の番号だけが MP(J) に登録されてい
て、最初の番号が登録されていないので、少し工夫が必要であるが、ソースを見れば分かると考える。

プログラム 6.1
DO 15 JJ=1,JP

DO 16 K=1,NFSHEL

IF(DABS(FT(K)-F(K,JJ)).GT.1.0D-5) GO TO 15

16 CONTINUE

C WRITE(6,*) ’ COIN ’,JJ,J,JP,NFSHEL

GO TO 19

15 CONTINUE

プログラム 6.1で示した部分が、TSBDFT が使っている一つの重要なアルゴリズムである。
DO 12 J=1,NSHEL (NSHEL は基底の総数）で基底関数の値を一つづつ計算していくのであるが、計算
された結果が、すべての代表点の k に対して等しくなっているものがあるので、それを取り除いている。
(8) であたえられる基底関数は、互いに同じ格子ベクトルを全く含んでいないのだから、当然互いに独立で
ある。しかし離散的な k－空間の代表点での値どうしはかならずしも独立ではない。大きさの小さい格子
ベクトルから順に行列の各列を計算していく時に新しく計算されたものが、既存の列の線形結合であるか
どうかの確かめが必要になる。ところが実際に計算してみると、この簡単な手続き、つまりすべての k に
対して同じになるものを除く手続きだけで独立な NFSHEL 個の基底関数を選び出せることが判明した。し
かもちょうど NFSHEL 個である。選び出された基底の番号が JPMAT(MAXKPT) に登録される。IPR を
4 か 5 にすると、小さな格子ベクトルのベースはすべて有効であるが、格子ベクトルが大きくなるにつれ
て無効のものの数がどんどん増えて、さんざん探した末に最後のベースが見つかるようすが再現される。
以上で NFSHEL 次元の正方行列ができるので、次は一次方程式を解いて係数を求めることになる。F と

FF と同じ行列を用意しているのは、汎用の一次方程式のサブルーチンが元の行列を変更ししまうからであ
る。あとで求められた係数を代入してあたえたエネルギー値に等しくなっているかを確かめている。
ここで使用している MLUDIS は NUMERICAL RECIPES の LUDCMP を中で CALL している。こ

れで LU 分解しておいて、各 k-点のエネルギー値をあたえて、それぞれのバンドの係数を決定していく。
MLULEQ は、おなじパッケージの LUBKSB を CALL している。ここのところは、使い慣れたパッケー
ジを使えばよいが、数百次元以上の行列を扱う必要があるので、実績と信用のあるパッケージを使う必要が
ある。
決定された係数 BB(J) を代入してあたえたエネルギー値が 10−13 以内で再現されていることを確かめ
て、B(MAXKPT,MAXEIG,2) に格納して次のサブルーチン FUEINT に引き渡す。
ここではエネルギー値を方程式の左辺に入れて、エネルギー値を求める係数を決めているが、ここで、

ENRRED で用意されている、s,p,d,f 成分を左辺にすれば、これらの成分値を求める係数が決定できる。

7 詳細な刻みでのバンドエネルギーの計算:FUEINT

6節の FNNMAT までの手続きは、すべての空間群で格子定数に無関係に使用でできるようになってい
る。しかしこの節以降で説明するサブルーチン、プログラムは空間群ごとに使い分ける必要がある。
この節では、立方格子、正方格子、直方格子で共通に使用できるものを紹介する。単純、面心、体心、底
心格子を共通に扱う関係で、結果を格納する領域と、計算手続きの両方に数倍の無駄をするルーチンになっ
ているが、最近の計算機事情を考慮して、プログラムの構造を単純にするほうに重点をおいたためである。
このパッケージに入れてある SUBROUTINE FUEINT は六方格子にも対応するようにしてある。この部
分は次の 8節で説明する。
このサブルーチンに 3節の FITPRM から引き渡される変数は
プログラム 7.1

COMMON/NXYZ/NX,NY,NZ,N64,NBLOCK,JS

の中に格納されている。NX,NY,NZ はそれぞれ a∗, b∗, c∗ 軸の刻み数で、N64 はそれらの最小公倍数であ
る。NBLOCK は今回のバージョンでは使わない。JS については表 1に示す。

表 1: 結晶点群と JS の値

No Point group JS

1 D2h, D2, C2v 1

2 C4h, C4, S4 1

3 D4h, D2d, C4v 2

4 Th, T 1

5 Oh, O, Td 3

6 D6h, C6v, D6, D3h 1

7 D3d, D3, C3v 1

7.1 領域の設定
空間群に、空間反転の対称性がなくても E(k) = E(−k) は時間反転が保証するので、E(k) の値を用意す
る k の範囲を決めるためには、空間群に空間反転を補って考えてもよい。表 1に各行に並べられた点群の
対称性を持つ結晶でははその行の最初にある点群として扱ってよいことになる。はじめの５個の点群は２
番の C4h を除き

kx = 0, ky = 0, kz = 0

の対称面を持っているので、負の値の k-座標を持つ E(k) はその座標の絶対値の座標での E と等しくなっ
ている。したがって第一象限に限ってエネルギー値を用意しておけばよいことになる。 C4h は kz = 0 の
対称面は持っているが kx = 0, ky = 0 の対称面を持っていないので、絶対値をとって第一象限に移すこと
はできないが、

kx < 0, ky < 0 −→ kx ← −kx, ky ← −ky (9)

kx < 0, ky > 0 −→ kx ← ky, ky ← −kx (10)

kx > 0, ky < 0 −→ kx ← −ky, ky ← −kz (11)

の手続きで第一象限にもってくることができる。10 節で説明するフェルミ面作図用の関数サブルーチン
FUNC にはこのための手当はしていない。動作テストができる実例があれば、簡単に手当ができると考え
ている。表 1 に含められている D6h, D3d については次節で説明する。

D4h では kx = ky の対称面があるので、kx ≥ ky の領域に制限できるので JS = 2 とする。Oh では、さ
らに kx ≥ ky ≥ kz と領域をせまくできるので、JS = 3 とする。Th では JS = 1 としている。これでは必
要な領域の３倍になっているが、簡単のために無駄を承知でこうしている。

TSBDFT では、この JS の値は利用者が自分であたえるようになっている。表 1にしたがって決めれば
よいのだが、上の説明を理解した上で利用されることを希望する。JS=1 が直方、JS=2 が正方、JS=3 が
立方対称と単純になっていなことを確かめ、その理由をなっとくした上で利用されたい。
単純格子ならば、kー空間の周期は a∗, b∗, c∗ である。したがって

0 ≤ kx ≤ 0.5a∗, 0 ≤ kx ≤ 0.5b∗, 0 ≤ kz ≤ 0.5c∗ (12)

の範囲のエネルギー値があれば、この範囲を出る k に対しては、a∗, b∗, c∗ を引く操作と、絶対値をとる操
作の組み合わせでこの範囲にいれることができる。(12) が D2h の場合の既約なブリルアンゾーンになって
いる。
底心格子では c∗ は kー空間の周期であるが、 a∗, b∗ は周期ではなく 2a∗, 2b∗ がこの方向の周期になる。

そこで
0 ≤ kx ≤ a∗, 0 ≤ kx ≤ b∗, 0 ≤ kz ≤ 0.5c∗ (13)

の範囲をとることにする。この範囲は既約なブリルアンゾーンの２倍の体積になっている。しかし、例えば

0 ≤ kx ≤ 0.5a∗, 0 ≤ ky ≤ b∗, 0 ≤ kz ≤ 0.5c∗ (14)

の範囲にして、既約なブリルアンゾーンと同体積にすることは可能である。しかし、10節で説明するフェ
ルミ面を描くための関数サブルーチン FUNC に

0.5a ≤ kx ≤ a∗, 0 ≤ ky ≤ b∗ (15)

の範囲にある k-点を
0 ≤ kx ≤ 0.5a∗, 0 ≤ ky ≤ b∗ (16)

の範囲に移し替える手続き

kx → a∗ − kx (17)

ky → b∗ − ky (18)

を入れておく必要が生じる。
面心格子や体心格子では 2a∗, 2b∗, 2c∗ がそれぞれの方向での周期になる。そこで

0 ≤ kx ≤ a∗, 0 ≤ kx ≤ b∗, 0 ≤ kz ≤ c∗ (19)

の範囲をとることにする。この範囲は既約なブリルアンゾーンの面心で２倍、体心で４倍である。面心直
方、体心直方、体心正方（面心正方）格子では上の底心直方格子の場合と似た手続きで既約なブリルアン
ゾーンと同体積の領域に還元することが可能である。また立方格子の場合ではいくつかのブロックに分割
することで、既約なブリルアンゾーンと同体積にすることは可能である。さきに説明を飛ばした NBLOCK
はこのことに対応するように残してある。

今回のバージョンで使われる領域は Th の体心格子で、既約なブリルアンゾーンの最高１２倍になってい
る。もちろんこれらに対応することは可能であるが、個別のケースでそれぞれに対応するほうが実際的で
ある。このバージョンではいろんな場合を一つのプログラムでカバーすることに重点を置いて、無駄の方に
は目をつぶっている。

FUEINT の中では、ここで述べた範囲は P(3) の中に記録されて、10 で述べる FUNC 用に fort.21 に出
力される。またこの範囲を刻んでいる刻みの数は N(3) で、やはり fort.21 に出力される。

7.2 エネルギー値の計算
計算の中心は (7) の ∑

α∈GP

cos(αTn · k) =
∑
m

cos(Tm · k) (20)

の基底関数の計算である。ここでの k は、入力であたえて最小公倍数 N64 を NL として

　 k = 2π(Ix/NL)a∗ + (Iy/NL)b∗ + (Iz/NL)c∗ (21)

と整数 Ix, Iy, Iz で一般に表すことができる。
一方 αTn = Tm は格子ベクトルであるから、格子の型で 1,2,3 と値を変える共通分母 Cd を用いて

　 Tm = (Lx/Cd)a + (Ly/Cd)b + (Lz/Cd)c (22)

とやはり整数 Lx, Ly, Lz で表すことができる。したがって (20) の各項は

　 cos(Tm · k) = cos(2π(IxLx + IyLy + IzLz)/(NLCd)) = cos(2π(IkT /(NLCd))) (23)

と整数 IkT を使って表される。
関数 cos は偶関数で、周期が 2π、さらに

cos(x) = − cos(π − x) = − cos(π + x) = cos(2π − x) (24)

だから、0 から π/2 の区間を NLCd/4 = Nq 分割した点で、cos の計算をしておけば必要なものは尽きるこ
とになる。FUEINT では、(24) を使って 0 から 2π の間の cos の値を CRK(0:600) に用意している。プロ
グラムの中では NLCd = KRMESH、Nq = MESH となっている。TSBDFT で、刻みの数 NX,NY,NZ
に加えてそれらの最小公倍数 N64 を入力として要求しているのは、刻み数を選ぶときになるべく最小公倍
数が小さくなることを使用者に意識してもらうためである。
これで準備が完了したので DO 21,DO 22,DO 23 でそれぞれ a∗, b∗, c∗ 方向を刻んで k を定める。このと

き JS の指定にしたがって、IY の上限 IYMM を IX にしたり、IZ の上限 IZMM を IY にしたりする。NN
は計算される k－点の番号で 10節 で説明する関数サブルーチン FUNC は３次元の k－空間の点にこの方
法で番号が振られていることを期待している。LATOM(3,MAXLP) に格納されている実格子のセットで与
えられている基底関数の中で JPMAT(MAXKPT) に登録されているものだけが、有効であることを注意し
ながら、DO 32 のループで基底関数に含まれている各格子ベクトルに対して (23) の IkT の計算とか、CRK
のデータの選択をして、それらの総和から基底関数値WW を計算する。これに B(MAXKPT,MAXEIG,2)
に格納されている FNNMAT で定めた係数を掛けて k でのエネルギー値が計算される。結果は 7.1 節で述
べた全体の見出し P,N,JS に続いて、各スピン、バンド毎に、最小公倍数 N64、k点の数 NN、スピンの向
き ISPIN、バンドの番号 NB の見出しにつづいて fort.21 に出力される。

8 六方格子
　

>a

b

>

b

>hex

ort

図 1: 六方格子と底心直方格子

この節では、表 1 に含められている六方格子を、底心直方格子として扱う方法を説明する。この方法を
適用すれば、10 節で説明するように、フェルミ面を作図するとき使う関数サブルーチン FUNC に直方、正
方、立方格子と同じものが使用できる利点がある。図 1 のように直方格子の b 軸を a 軸に垂直に

√
3a の

長さに取る。つまり b 軸の長さが a 軸の長さのちょうど
√

3 倍の底心直方格子は六方格子である。

*

b* he
x

a*hex

ortb*ort a

v
vv

v

図 2: 逆格子での六方格子と底心直方格子

六方格子と底心直方格子の対応を逆格子で見たのが、図 2 である。六方格子では a∗ 軸が a 軸に３０度
傾いて、その大きさが

|a∗hex| =
2√
3|a| (25)

になるのに対し、直方格子では a∗ 軸が a 軸の方向で、大きさが

|a∗ort| =
1
|a| (26)

となる。したがって長さの関係は
|a∗hex| =

2√
3
|a∗ort| (27)

となり、たがいに 30度傾いている。また

|b∗ort| =
1√
3
|aort|∗ (28)

である。
D6h の場合の既約なブリルアンゾーンは図 2 に多く見えている直角三角形で、赤い矩形で示した領域は
６倍の体積である。D3d では、TSP:p67～68 に説明してあるように P31m と P3m1 の２種類があるが、
赤い矩形で示した領域は共に３倍の体積で既約なブリルアンゾーンをカバーしている。
フェルミ面を作図するとき使う関数サブルーチン FUNCに渡す関数値は、直方格子の a∗ort, b

∗
ortをNX, NY

個に刻んだ点の関数値であるが、(7) の k は a∗hex, b∗hex を基準にしたものである。両者の変換は同じ k を
両方の基準で表して、図 2 から関係を求めると

　 k = xa∗ort + yb∗ort = Xa∗hex + Y b∗hex (29)

X = (x− y)/2 (30)

Y = (x + y)/2 (31)

となる。プログラム 8.1はサブルーチン FUEINT からの関係部分の抜粋である。IZ の部分に ∗2 としてあ
るのは、(30)、(31) にある /2 を 分母の KRMESHに含めてあるが、これは kz には余分になるので、こ
こで調整している。

プログラム 8.1
IF(IL.GE.1) THEN

IRK=LATOM(1,I)*IX*MPX+LATOM(2,I)*IY*MPY

& +LATOM(3,I)*IZ*MPZ

ELSE IF(IL.EQ.0) THEN

IRK=LATOM(1,I)*(IX*MPX-IY*MPY)

& +LATOM(2,I)*(IX*MPX+IY*MPY)

& +LATOM(3,I)*IZ*MPZ*2

END IF

9 詳細な刻みのエネルギーを用いた状態密度:SDOSDR

ほとんどのバンド計算のプログラムは、フェルミエネルギーを出力としてあたえているが、7節で FUEINT
がより詳細な刻みでエネルギーがあたえているので、あらためてフェルミエネルギーを決定し直しておく
必要がある。このために SDOSDR というプログラムが用意されている。四面体法で各バンド毎の状態密
度とそのエネルギー積分を計算し、それに基づいてフェルミエネルギー決定する。状態密度とその計算法で
ある四面体法については別に解説を用意しているので、詳しくはそちらに譲ることにして SDOSDR の機
能を簡単に述べる。7 節で説明した内挿エネルギーのファイル fort.21 から情報を読み取って計算を実行す
る。この計算には空間群や格子定数の情報は直接には必要としない。

入力データ 9.1
-0.22, 0.0001, 9501, 5.5 E0,DE,NNEE,ELECN

入力データ 9.1 は SDOSDR 用の入力データの Cu の例で E0 から、刻み DE で NNEE 個のエネルギー
点で状態密度の計算を実行する。最後の ELECN はバンドに含まれる電子数であるが、使い方に注意が必
要である。すべてのバンドは、単位包あたり１個の電子と規格化されている。Cu の場合６枚のバンドの内
挿結果が fort.21 にあたえられている。この場合常磁性で反転対称がある結晶なので１枚のバンドに２個の
電子を収容する。１１個の電子がこのバンドに収容されているので、半分にして 5.5 としてある。

入力データ 9.2
0.30D0 ,0.0001D0, 7000, 6.0D0

入力データ 9.2 は Ni のためのものである。ここでは３番から６番のバンドがフェルミ面を持つことが分
かっているので、この部分だけの内挿値が fort.21 に格納されている。１０個のバレンス電子から１番２番
のバンドの UP と DOWN の電子の分 4 を引いて 6.0 としてある。

10 逆格子空間の任意の点のエネルギー:FUNC

この節では、11 節で説明するフェルミ面の作図用の TPERSP のプログラムの中に含まれている関数サ
ブルーチン FUNC の中で使われているアルゴリズムを説明する。

プログラム 10.1
FUNCTION FUNC(XX)

IMPLICIT REAL*8 (A-H,O-Z)

REAL*8 KB

COMMON/ENRSPL/EM(1900000),PN(3),P(3),N(3),IS

DIMENSION IFAC(3)

DIMENSION A(4),IPER(3,6),FF(2,2,2),F(8),KB(3)

EQUIVALENCE(F(1),FF(1,1,1))

DATA IPER/1,2,3, 1,3,2, 3,1,2, 2,1,3, 2,3,1, 3,2,1/

DATA IFAC/1,2,4/

TPERSP の説明の pdf の４節の隠線処理のアルゴリズムで説明しているように、FUNC には隠線処理
のため繰り返し、不特定の連続した点での値が要求される。TPERSP が使用しているこの隠線処理のアル
ゴリズムは、現在の観点からすれば、効率的なものとはいい難いが、任意の箇所を切り取ることを可能にし
ているなどの機能は捨て難い。現在の計算機の能力が、この非効率さを十分にカバーしている。

FUNC には繰り返し、しかも接近した点での値を要求されることに対応した工夫が盛り込まれいる。し
かし２０年以上も前に作られたものであるため、当時の事情を反映しており、現在では少し分かりづらいも
のになっている。以下に筆者自身が昔の自作のプログラムを解読した結果を紹介する。プログラム 10.1は
FUNC の最初にある宣言文の部分であるが、ここに表れている IPER(3,6) と IFAC(3) の整数データの意
味の説明に注意されたい。

10.1 四面体利用線型内挿
アルゴリズムの基本は、図 3 に示された四面体の 4個の角の値から

E = A1x + A2y + A3z + A4 (32)

の一次式の係数を決定して、四面体内部の任意の点での関数値を求める線型内挿である。ここで x, y, z の
原点は１番の角にとることにする。したがって、 A4 は１番の角の関数値そのものである。
図 3の直方体は、7節で説明した FUEINT が計算して FUNC に値をわたした、細かいメッシュの一つ
を現している。辺の数字は四面体につけた番号で、直方体の角には小さめの数字で示した番号がついてい

図 3: 直方体の６個の四面体への分割。

る。図 3の６個の四面体は、どれも１番の角と８番の角を結ぶ体対角線を、共通の辺として持っている。１
番の角と８番の角はすべての四面体で共通である。さらに面の対角線は、各四面体が 2本づつ持っている
が、隣り合う二つの四面体が共有している。四面体の辺の残り３本は直方体の辺である。
図 3で赤い線で示した１番の四面体を見てみよう。ここでの直方体の辺に沿って１番の角から８番の角へ
向かう道筋は、最初 x方向で、次が y方向で、最後が z方向である。この y方向へ向かう辺はこの四面体
だけが占有している。この四面体での、(32)の係数 A1 は２番の角の関数値 E2 から１番の角の関数値 E1

を引いたもので決まる。両者で y座標と z座標が共通であるからである。同様に A2 は４番の角の関数値
E4 から２番の角の関数値 E2 を引いたもので決まる。また A3 は８番の角の関数値 E8 から４番の角の関
数値 E4 を引いたもので決まる。
ここで、プログラム 10.1 の IPER の数値と図 3の四面体の番号を見比べてみよう。IPER の数値 1,2,3
を x,y,zと対応させてみると、６組の IPER が図の四面体の番号と対応していることが見て取れる。１番の
四面体は上に説明したように、x,y,z つまり 1,2,3 で、２番は、x,z,y(1,3,2) である。以下同様であるが念の
ため、たとえば５番をみてみると、y,z,x(2,3,1) となっている。IFAC(3)/1,2,4/ はそれぞれ、x,y,z 方向に移
動したときに角の番号の変化に対応している。つまり x方向では１番だけ変わり、y方向ではちょうど２番
だけ、z方向ではいつも４番だけ変わっている。

プログラム 10.2
C *** DETERMINE THE COEFFICIENTS OF LINEAR FORM ***

K1=IPER(1,IP)

K2=IPER(2,IP)

K3=IPER(3,IP)

I1=IFAC(K1)

I2=IFAC(K2)

I3=IFAC(K3)

A(4)=F(1)

JA=I1+1

A(K1)=(F(JA)-F(1))/P(K1)

JB=I1+I2+1

A(K2)=(F(JB)-F(JA))/P(K2)

A(K3)=(F(8)-F(JB))/P(K3)

プログラム 10.2の IP が x,y,zの大小関係から決めた四面体の番号である。ここで、I3 を求めているが使
用していない。最後はどの四面体でも F(8) になると決まっているからである。

10.2 計算されている領域への座標変換
逆格子空間の中で 7.1 節で説明した領域の中だけで、関数値が用意されているので、逆格子空間の中の
任意の点を、この領域内に系の対称性に即して変換することが FUNC の最初の手続きになる。プログラム
10.3 がこの部分にあたっている。XX(3) は関数値が要求された座標で、これを KB(3) に置き換えて変換に
とりかかる。もちろん、この KB は REAL*8 KB と宣言してあって、実数変数である。DO 1 のループで
D2h の場合の変換が完了する。7.1 節で、P(3) としていた変数がここでは PN(3) になっている。PN(1) は
単純格子では 0.5 で、他の格子では 1.0 になっている。つまり、今の座標系は、直交座標ではあるが、そ
の単位は三方向でそれぞれ a∗, b∗, c∗ となっている。
このバージョンでは、C4h の場合の手当がしていないが、(9)、(10)、(11) であたえられている変換の手
続きをこの部分に加えればよいことになる。また (17)、(18) の変換をこの部分に加えれば底心直方格子で
の関数値を用意する範囲を半分にすることができる。

7 節で JS としていた変数がここでは IS となっている。IS=1 なら GO TO 4 となっている。プログラ
ム 10.3 の最後の行は、プログラム 10.4 の先頭と同じ行である。これで座標変換の手続きは終了して、3 図
で示されている、どの直方体の中にいるかを見つける次の手続きに移る。IS=2、IS=3 なら KB(3) の３個
の値の大小関係をみて、データを入れ替える変換をしている。IS=3 で KB(2) と KB(3) を入れ替えたら、
KB(1) と KB(2) の大小関係を再度チェックする必要がある。

プログラム 10.3
DO 1 I=1,3

KB(I)=XX(I)

2 KB(I)=ABS(KB(I))

IF(KB(I).LE.PN(I)) GO TO 1

KB(I)=KB(I)-PN(I)*2.0

GO TO 2

1 CONTINUE

IF(IS.EQ.1) GO TO 4

C *** KB(1).GE.KB(2).GE.KB(3) ***

IF(KB(1).GE.KB(2)) GO TO 3

W=KB(1)

KB(1)=KB(2)

KB(2)=W

3 CONTINUE

IF(IS.EQ.2) GO TO 4

IF(KB(2).GE.KB(3)) GO TO 4

W=KB(2)

KB(2)=KB(3)

KB(3)=W

IF(KB(1).GE.KB(2)) GO TO 4

W=KB(1)

KB(1)=KB(2)

KB(2)=W

4 IX=KB(1)/P(1)

10.3 直方体の角の関数値を決定する
プログラム 10.3 で、関数値が用意された領域に送り込まれた KB(3) の点がどこにあるかを次にみつけ
ることになる。この部分がプログラム 10.4 に示されている。まず KB(3) を三つの方向の刻み巾 P(3) で
割って整数値 IX、IY、IZ を求める。これらの値がちょうど端の値 N(3) になっているときは、一つ手前に
もどしておく。
プログラム 10.4

4 IX=KB(1)/P(1)

IF(IX.EQ.N(1)) IX=N(1)-1

IY=KB(2)/P(2)

IF(IY.EQ.N(2)) IY=N(2)-1

IZ=KB(3)/P(3)

IF(IZ.EQ.N(3)) IZ=N(3)-1

IF(IS.EQ.1) NI=(IX*(N(2)+1)+IY)*(N(3)+1)+IZ+1

IF(IS.EQ.2) NI=((IX*(IX+1))/2+IY)*(N(3)+1)+IZ+1

IF(IS.EQ.3) NI=(IX*(IX+1)*(IX+2))/6

1 +(IY*(IY+1))/2+IZ+1

C *** IF THE BLOCK IS THE SAME, GO TO 40 ***

IF(NI.EQ.NIOLD) GO TO 40

C *** PICK UP THE ENERGIES OF 8 POINTS ***

NK=NI

IF(IS.EQ.1) MX=(N(2)+1)*(N(3)+1)

IF(IS.EQ.2) MX=(IX+1)*(N(3)+1)

IF(IS.EQ.3) MX=((IX+1)*(IX+2))/2

DO 41 KX=1,2

DO 42 KY=1,2

DO 43 KZ=1,2

FF(KX,KY,KZ)=EM(NK)

NK=NK+1

43 CONTINUE

IF(IS.NE.3) NK=NK-2+(N(3)+1)

IF(IS.EQ.3) NK=NK-2+IY+KY

42 CONTINUE

IF(KX.LT.2) NK=NI+MX

41 CONTINUE

40 CONTINUE

関数値が格納されている EM(NI) は添字が一つの配列であるので、IX、IY、IZ から NI を求めなければ
ならない。IS=1,2,3 のそれぞれについて、IX、IY、IZ から NI を求めている式は、7節 で説明した、サブ
ルーチン FUEINT の DO 21 IX=,DO 22 IY=,DO 23 IZ= の順序と JS（この FUNC では IS) の扱い方に
したがって、周知の数列の和の公式から簡単に導くことができる。この NI が図 3 の 1 番の角に相当する。
次に IF(NI.EQ.NIOLD) GO TO 40 としてあるのが FUNC が接近した座標値にたいして値が要求される
ことに対応した、応答を速くする一つの工夫である。図 3 の 他の７つ角の NK(NI) を求めるのは、それぞ
れの IS に対する式で IX,IY,IZ の１だけの違いが NI にどのように影響するかをみることで簡単にもとま
る。結果を FF(KX,KY,KZ)=EM(NK) と格納して、プログラム 10.2 で F(1～8) で使用している。この使
い分けのために、プログラム 10.1 で EQUIVALENCE(F(1),FF(1,1,1)) と宣言して二つの配列に同じアド
レスをあてがってある。FF の KX,KY,KZ の順序が 3 図の角の番号をあたえている。これは配列内のアド
レスに対する FORTRAN の規約にしたがっている。このプログラムで、たとえば IS=2 のとき、IX=IY
の直方体の場合に３番と７番の角の NK が不自然な値になり、したがってこの角の関数値がおかしな値に
なるがこれは気にしなくてよい。この関数値が使われる４番、５番、６番の四面体が選ばれることがないか
らである。

10.4 四面体の番号の決定
プログラム 10.5

C *** FIND THE TETRAHEDRON ***

X=KB(1)-P(1)*FLOAT(IX)

Y=KB(2)-P(2)*FLOAT(IY)

Z=KB(3)-P(3)*FLOAT(IZ)

IF(X.GE.Y) GO TO 11

IF(Y.GE.Z) GO TO 12

C WRITE(6,*) X,Y,Z

IP=6

GO TO 13

12 IF(X.GE.Z) GO TO 24

IP=5

GO TO 13

24 IP=4

GO TO 13

11 IF(Y.GE.Z) GO TO 21

IF(X.GE.Z) GO TO 22

IP=3

GO TO 13

22 IP=2

GO TO 13

21 IP=1

C *** IF THE TETRAHEDRON IS THE SAME, GO TO 10 ***

13 IF(NIOLD.NE.NI) GO TO 14

IF(IPOLD.EQ.IP) GO TO 10

14 IPOLD=IP

NIOLD=NI

つぎに、プログラム 10.5 で 3図 の 1 番の角が原点になるようにして、X,Y,Z を求めてそれらの大小関
係から、四面体の番号 IP を決定して、順序を変えて最初に説明したプログラム 10.2 にはいる。

IF(IPOLD.EQ.IP) GO TO 10 で係数の決定をとばしているのは、やはり応答を速くするための工夫で
ある。処理系によっては、これに関連した変数 A,F,IPER,IFAC の SAVE 宣言が必要になる。

11 フェルミ面の実例
11.1 フェルミ面作図プログラム fermhcs.f

この節では 7 節、8 節で説明した FUEINT を含んだ TSBDFT の出力 fort.21 を用いて、立方、正方、直
方、六方 格子に共通に使える TPERSP のフェルミ面作図プログラム fermhcs.f を紹介する。このプログラ
ムではブリルアンゾーンの外側を切り取るほか、フェルミ面の手前上方の８分の１を切り取っている。この
プログラムは TPERSP の説明の６節で説明しているものとほとんど同じである。以前のものが面心立方格
子の Cu の６番目のバンドに特化しているのに対し、このプログラムは六方格子を含む、広範な物質に対応
している。以下は変更点を中心にした説明である。なお六方格子関連の部分はプログラムでは IF(IL.EQ.0)
と明確に読み取れる。説明を分かりやすくするため 11.5 節の PtBi の項で行う。
入力データ 11.1
6,1 NB,ISPIN

80.0 SIZE

1 IEOH 1,elctron -1 hole

0.4157 EF

129 129 129 1 1 1 NN,NK

15.0 ALPHA

各種の状況に対応するため、標準入力からの入力が必要になる。入力データ 11.1 は図 4 のためのデータ
である。横につけたコメント文から内容は読み取れる。TPERSP の引数から ALPHA(AF) だけを外から
変更できるようにしてあるのは、この値が図の見易さに大きく影響するからである。
図の範囲と大きさを決める TPERSP の引数 XFU と WF をプログラム 11.1 のようにしている。これ
で a∗, b∗, c∗ の大きさをそれぞれの方向の単位にし、図の寸法も実際の比になっている。つまり直方格子や、

正方格子では３方向の倍率が異なっている。このバージョンでは扱っていない単斜格子や三斜格子を含めて
扱うためには、TPERSP の説明の２節で述べているように３方向とも a∗ を単位にした直交座標を採用し
て TSPACE のサブルーチン RECTAG と GETKVC を活用するのがよいかもしれない。
プログラム 11.1

XFU(1,1)=-PN(1)

XFU(2,1)=-PN(2)

XFU(3,1)=-PN(3)

XFU(1,2)=PN(1)

XFU(2,2)=PN(2)

XFU(3,2)=PN(3)

READ(5,*) SIZE

WASIZE=XFU(1,2)-XFU(1,1)

WF(1)=SIZE

WF(2)=SIZE*(A/B)*(XFU(2,2)-XFU(2,1))/WASIZE

WF(3)=SIZE*(A/C)*(XFU(3,2)-XFU(3,1))/WASIZE

TSPACE のサブルーチン TSBZEG があたえるブリルアンゾーンエッジの座標は３方向で共通に a∗ を
単位にしているので、単位の換算を行っている。
ブリルアンゾーンの外側を切り取る図には切り取った切り口の手当が必要である。Cu ではブリルアン
ゾーンの L 点に切り口があることが自明なので、８個の L 点を中心にして、それぞれの方向の面で切った
切り口の TPSECT を CALL していた。しかし一般にはどのブリルアンゾーンの面に切り口があるかは予
想がつかないのですべての面で TPSECT を CALL する。TSBGEG の引数 NRECPT は面を与えている
逆格子ベクトルを a∗, b∗, c∗ 単位であたえているので、面の方向をその方向にし、原点との中点を中心にし
て TPSECT を CALL すればよいことになる。ただし図の範囲を指定する面と一致することもあるので、
TPFXYZ の方のこの部分の引数を 0 にして切り口の図が重ならないようにする必要がある。　

11.2 Cu

(a) (b) (c)

図 4: Cu のフェルミ面

図 4 の３個のフェルミ面のは左から、19 個の代表点、85 個の代表点、489 個の代表点から TSBDFT で
内挿した結果を用いた図である。注意して見ないと違いが分からないが、(a) 図ではほぼ円形であるはずの
胴回りの切り口にわずかな波打ちがみられる。(b) と (c) とでは、(b) の下部に少し不自然な凹凸がみられ
る。この凹凸が (c) にはみられないので、不自然としているが、(c) がなければ判定の基準はない。代表点を
どの密度にとるべきかは一般的には決められない。どの程度の細かい構造を議論するのかに依存している。

11.3 Ni

強磁性金属の例として Ni を取り上げる。AYBAND の 4.4 節の Ni のバンド図と見比べながら説明を読
んでいただきたい。Ni では fermhcs.f の手前上部８分の１を切り取る機能をおさえた図を示す。

(a) (b)

図 5: Ni 下向きスピンの第３バンド (a) と第４バンド (b) のホール面

図 5 は X 点の周りの小さな二つのホール面である。それぞれ 0.0046 個と、0.0134 個のホールを含んで
いる。AYBAND のバンド図では ∆ 軸では両者が縮退しているが S 軸と Z 軸では離れている。

図 6: Ni 下向きスピンの第 5バンドの電子面

図 6 は下向きスピンの第５バンドのフェルミ面で電子面として描いている。0.524 個の電子を含んでいる
のでホール面としたほうが体積が小さくなるが、図に示すように電子面として閉じている。
図 7 は第６バンド二つのスピンの向きの電子面を示している。それぞれ 0.179 個と 0.310 個の電子を含ん

でいる。AYBAND の図では ∆ 軸で複雑な偶然縮退が起きているのが示されているが、ここに示したフェ
ルミ面には、不自然さはあまり現れていない。

(a) (b)

図 7: Ni の第 6バンドの下向きスピン (a)と上向きスピン (b)の電子面

11.4 Mo

体心立方格子の例として Mo を取り上げる。遷移金属で 5s 電子２個と、4d 電子４個の合計６個の価電
子を持っている。図 8 の (a) が計算されたバンド図である。常磁性で反転対称を持つので、バンド１枚あ
たり２個の電子を収容する。

図 8: Mo のバンド図 (a)と第５バンドの電子ポケット (b)

図 8 (a) のバンド図には、フェルミエネルギーと交わるバンドに、上から順に青、緑、赤の色がつけてあ
る。この手法はバンド図からフェルミ面の様子を推測するのに有効である。この物質のスピン軌道相互作用
はあまり強くないので、それを入れないバンドの偶然縮退が僅かに解けた結果の急な折れ曲がりが多く見
られる。また ∆ 軸、 Λ 軸、F 軸では入れないバンドの二重縮退が僅かに解けているが、図のスケールで
は分離して見れない。F 軸から Λ 軸にかけて、青の線と緑の線が重なって見えるがじっさいは僅かに分裂
している。計算に使われた代表点は Γ 点と H 点の間を 16 分割した 285 点であった。この密度でバンドの
急な変化を大体表すことができている。
青の第５バンドは、ほとんどフェルミエネルギーの上を走っているが、∆ 軸の一部で切っている。この
部分のフェルミ面が図 8 (b) で含まれる電子数は 0.009 である。

(a) (b)

図 9: Mo の第４バンドの電子面 (a)と第３バンドのホール面 (b)

緑の第４バンドは Γ 点の周りと ∆ 軸にかけてフェルミエネルギーの下になっている。∆ 軸では青のバ
ンドとの間で、上に述べた偶然縮退のスピン軌道相互作用による分裂が見えていてここでバンドの乗り移
りが起きている。図 9 (a) では、これに対応して ∆ 軸のまわりにフェルミ面の膨らみがある。この面に含
まれる電子数は 0.191 個である。
赤の第３バンドは H 点と N 点の周りだけでフェルミエネルギーの上に出ている。図 9 (b) はこの部分を

ホール面として描いている。H 点の周りは図では６個に切れているが、これらは集まって一つの正八面体
型の閉じた面になる。また N 点の周りは向かい合った面のものがつながって、６個の楕円面を形成する。
これら 2種類のフェルミ面に含まれるホール数は 0.200 個である。

11.5 PtBi

図 10: PtBi の結晶構造とブリルアンゾーン

六方格子の例として PtBi を取り上げる。図 10 に示す PtBi の結晶は空間群 D4
6h, P63/mmc を持ってい

る。したがってスピンー軌道相互作用を入れないバンドでは kz = c∗/2 のブリルアンゾーン面で２重縮退
している。スピンー軌道相互作用は R 軸を除いてこの縮退を解くがあまり大きくはない。

8 節で述べたように、底心直方格子としてエネルギー値があたえられているが、TSPACE には六方格子
としてデータが蓄えられている。したがって、fermhcs.f には５箇所に IF(IL.EQ.0) の IF 分が挿入されて、

図 11: PtBi のバンド図 (a)と第１４バンドのホール面

六方格子に対応している。

プログラム 11.2
IF(IL.EQ.0) B=SQRT(3.0D0)*A

プログラム 11.2 が最初の挿入で、これで六方格子を TSPACE の持っている格子定数から切り離して、こ
のプログラムでは　 B =

√
3A の底心直方格子として扱うことになる。

次の挿入はブリルアンゾーンを描く部分である。TSBZEG があたえるブリルアンゾーンエッジの座標
CO(3,2,100) は、a∗ を単位にした直行座標であたえられているが、ここでは直交する三方向でそれぞれ、
a∗, b∗, c∗ を単位にしているので、六方格子以外ではスケール変換をして TPLINE に渡している。六方格子
では CO のデータを TSPACE のサブルーチン GETKVC を使って、いったん六方格子の a∗hex, b∗hex, c∗hex

を単位にした斜交座標の値 AK1,BK1,CK1,AK2,BK2,CK2 に変換している。つぎに (30), (31) を逆に解い
た式 x = X + Y, y = X − Y で TPLINE に渡す値を求めている。次はブリルアンゾーンの外側を切り取っ
たときの切り口に TPSECT を使って蓋をしている部分である。TSBZEG の出力の一つ NRECPT(3,30)
はブリルアンゾーンを形成する面に対応する逆格子を表している。通常の格子なら今使っている座標系そ
のものであるから、そのまま面の方向余弦とし、中点を面が通る点にしている。六方格子では a∗hex, b∗hex 系
での値を底心直方格子のそれに変換している。
あと２箇所は FUNCD と LFUNC の中で、あたえられた点がブリルアンゾーンの中か外かを判定してい
るところである。TPERSP が使っている底心直方格子の座標を TSPACE が使用している六方格子の座標
に (30),(31) を用いて変換をしている。
図 11 (a) の下にある２枚のバンドは Bi の 6s－バンドで、その上に Pt の 5dーバンドが１０枚見えて
いる。赤色で示した１４バンドは Rー軸では下の１３バンドと縮退している。このバンドは Γ 点の周りで
フェルミエネルギーの上に出ている。図 11 (b) はこの部分をホール面として描いている。この面に囲まれ
た体積はブリルアンゾーンの体積の 0.024 にあたり 0.048 個のホールを含んでいる。
図 11 (a) でマゼンダ色で示した第１５バンドは、Rー軸で上の第１６バンドと縮退しているので図では
青色にしている。このバンドは Γ 点から Λ 軸の端の A点まで、さらに Rー軸全体の A点から L点まで

(a) (b)

図 12: PtBi の第１５バンドのホール面 (a)と第１６バンドの電子面

がフェルミエネルギーの上になっている。この１５バンドのフェルミ面をホール面として描いたのが図 12
(a) である。R軸に沿った６本の腕がフェルミ面の６回対称性をよく表している。8 節で導入した六方格子
を B =

√
3A の底心直方格子として扱う方法が、よく機能していることを示している。このフェルミ面に

囲まれた体積は 0.277 である。
図 11 (a) でシアン色で示した第１６バンドは、H点で最も低いエネルギーになり、P軸は端のK点まで

がフェルミエネルギーの下になっている。K点から T ′ 軸に沿って M点に向かってエネルギーが下がって
行く。この１６バンドのフェルミ面を電子面として描いたのが 12 (b) である。kz = c∗/2 の面上では R軸
で１５バンド縮退しているだけでなく S, S′ 軸でも１５バンドと１６バンドが接近しているのが図 11 (a)
に示されているが、このことは図 11 (a) と図 11 (b) がこの面の上で互いに相手の隙間に広がっていること
にも表れている。
ここで使用している fermfcs.f のフェルミ面の手前上方８分の１切り取る機能が、図 11 で特徴的な二種
類の断面を示している。この格子では c∗ 面には６０度の周期があるので、９０度異なる面がちょうど３０
度だけ異なる面に対応することになる。

11.6 Ga

　最後に単体の金属 Ga をとりあげる。この物質は TPERSP と AYBAND の両方の解説でとりあげて
いる。結晶構造と空間群の情報は TPERSP を、ブリルアンゾーンとバンド構造は AYBAND を参照され
たい。

AYBAND の説明で触れたように、半金属的なバンド構造でのホール面は対称軸からはずれた箇所に存
在している。AYBAND で使用した代表点は 115 点であったが、その内５点だけがこのバンドでフェルミ
エネルギーの上であった。これでは明らかに不十分な代表点の数と予想される。そこで代表点数を 322 点
にして、NX,NY,NZ を 192,96,192 にして、TSBDFT を使用して 1,815,937 点の詳細メッシュでフェルミ
エネルギーを、SDOSDR で決め直した結果 AYBAND の図３、図４で示したバンド図とは少し変わって、
ホールバンドが B軸を切るように変わった。この軸は２重縮退であるので、すぐ下のバンドも小さいなが
らフェルミ面を持つことになる。
上の内挿結果を用いたフェルミ面が図 13の (a)図である。ブリルアンゾーンが六角柱になっているが、六

(a) (b)

図 13: Ga のホール面 代表点 322(a)と 2119(b)

方格子のそれとは異なって６回対称があるわけではない。ブリルアンゾーンの軸や対称点の名称はTSP:P265
の図 A.18 か、AYBAND の図２を参照されたい。直方対称を反映して、kx = 0 の面を対称面にする大き
めの塊と、少し小さめの ky = 0 の面を対称面にする塊がある。kz = 0 の面に小さな楕円面 (e1)と細い
ネック (n1)が見えている。kz = c∗/2 のゾーン境界のネック (n2)は B軸の上にある。もう一つの対称面
ky = b∗/2 の面では少し太いめの膨らみで隣のゾーンと繋がっているが、その手前の対称面から外れたとこ
ろに細いネック (n3)がある。さらに上下の大きめの塊をつないで kz = 0 の面に細いネック (n1)を持った
上下に伸びた瓢箪型の面と大きめの塊との繋がりも細いネック (n4)になっている。
これらの細かい構造が真実かどうかを確かめるために、さらに代表点の数を 2119 にして描いた図が図 13
の (b) 図である。ここでは (e1)が大きくなって、(n1)と合体している。B軸上のネック (n2) は大体同じで
ある。(n3) は太くなってネックを形成していない。(n4) は (a)図より細くなっているが、繋がっているこ
とは確認できる。
代表点が多いのだから (b) 図のほうが確からしいとは言えるが、これが真実であるとは結論できない。確

かめるためにはさらに代表点を増やしてバンド計算を実行する必要がある。半金属であることを反映して
か 322 点で決めた自己無撞着ポテンシャルは 2119点でも十分に収斂したポテンシャルであった。したがっ
てより大きな代表点を用いての、自己無撞着計算は不要なので実行することは可能である。しかしそれは
本稿の目的を超えていると考える。一般に半金属のフェルミ面はこの Ga のように対称軸を外れたところ
にあることが多く、しかも形が小さいためブリルアンゾーン全体のスケールからみれば、バンドエネルギー
の細かな変化がフェルミ面の形に表れる。

AYBAND の図４で青で示したバンドの電子面が図 14 である。(b) 図には代表点の数に伴って、(a) 図よ
りも細かな構造が見えている。T 点の周りに B 軸と H 軸に沿った塊が見えている。図では４個に分かれて
いるが、ブリルアンゾーンの面で切られているためで、実際は一つにまとまって閉じたフェルミ面になる。
Σ 軸の周りにあるフェルミ面がブリルアンゾーンの面で切られた切れ端がブリルアンゾーンの縦のエッジが
C 軸と交わるところに見えている。この面は右と左に同じものがあるが、互いに繋がっている訳ではない。

AYBAND の図４で緑で示したバンドの電子面が図 15 である。ここでは代表点の数の違いはほとんど見
られない。この面は図 14 と B 軸で繋がっている。楕円面のように見えているが、本当は B 軸との交点の
先端は尖っている。

(a) (b)

図 14: Ga の電子面 (1) 代表点 322(a)と 2119(b)

(a) (b)

図 15: Ga の電子面 (2) 代表点 322(a)と 2119(b)

